10 Processamento de sinais

Problema 3
Uma resistência de 3 kΩ e um condensador de 5 nF estão ligados em série a uma fonte com tensão $V_\text{e}(t) = 2 - 2\,t$, entre $t = 0$ e $t = 4$, e $V_\text{e}(t) = 0$ nos outros instantes (t medido em μs e V_e em V). Calcule a corrente no circuito em $t > 0$.

Convém primeiro definir o sistema de unidades a usar:

$$1\,\Omega = \frac{1}{\text{F}\cdot\text{Hz}} \quad \Longrightarrow \quad 1\,\text{k}\Omega = \frac{1}{\text{nF}\cdot\text{MHz}}$$

Como tal, pode usar-se kΩ para a resistência, nF para a capacidade e a frequência s estará em MHz. A impedância equivalente da resistência em série com o condensador é:

```
(%i1) z: 3 + 1/(5*s)$
```

A unidade para o tempo será a inversa da unidade da frequência, ou seja, μs e não é necessário alterar a expressão da tensão. A tensão da fonte, em função do tempo, pode escrever-se da forma seguinte:

$$V_\text{e}(t) = (2 - 2\,t)\,(1 - u(t-4)) = 2 - 2\,t + u(t-4)\,(6 + 2(t-4))$$

Ou seja, é a sobreposição de uma tensão $V_1 = 2 - 2\,t$ mais outra tensão $V_2 = 6 + 2\,t$ deslocada 4 unidades em t. Calculam-se as correntes I_1 e I_2 produzidas por cada uma dessas tensões:

```
(%i2) V1: 2 - 2*t$

(%i3) i1: ratsimp(laplace (V1,t,s)/z);
```

(%o3) $$\frac{10\,s - 10}{15\,s^2 + s}$$

```
(%i4) I1: ilt(i1,s,t);
```

(%o4) $$\frac{32\,\mathrm{e}^{-\frac{t}{15}}}{3} - 10$$

```
(%i5) V2: 6 + 2*t$

(%i6) i2: ratsimp(laplace(V2,t,s)/z);
```

(%o6) $\dfrac{30\,s+10}{15\,s^2+s}$

```
(%i7) I2: ilt(i2,s,t);
```

(%o7) $10-8\,\mathrm{e}^{-\frac{t}{15}}$

A corrente (em mA, t em µs) é $I_1(t)$ mais $I_2(t)$ deslocada 4 unidades em t:

$$I(t) = I_1(t) + u(t-4)\,I_2(t-4) = \frac{32}{3}\mathrm{e}^{-t/15} - 10 + u(t-4)\left(10-8\mathrm{e}^{-(t-4)/15}\right)$$

Problema 6
No circuito da figura: (a) Calcule a impedância total, em função de s. (b) Calcule a transformada da corrente que passa pelo indutor. (c) Encontre a função de transferência, se a tensão de saída for medida no condensador. (d) Determine a equação diferencial para a tensão de saída.



(a) O condensador e a resistência estão em paralelo e esse sistema está em série com o indutor; como tal, a impedância total é:

```
(%i8) z: ratsimp (L*s + R/(C*s)/(R + 1/(C*s)));
```

(%o8) $\dfrac{\left(s^2\,C\,L+1\right)\,R+s\,L}{s\,C\,R+1}$

(b) Representando a transformada de Laplace $\widetilde{V}_\mathrm{e}$ da tensão de entrada com a variável ve, a transformada da corrente total (no indutor) é:

```
(%i9) i: ve/z;
```

(%o9) $\dfrac{ve\,(s\,C\,R+1)}{\left(s^2\,C\,L+1\right)\,R+s\,L}$

(c) A tensão no condensador é a mesma do que no sistema do condensador em paralelo com a resistência; como tal, a transformada da tensão no condensador é:

```
(%i10) v: ratsimp (i*R/(C*s)/(R + 1/(C*s)));
```

(%o10) $$\frac{ve\,R}{\left(s^2\,C\,L+1\right)\,R+s\,L}$$

E a função de transferência é:

```
(%i11) h: v/ve;
```

(%o11) $$\frac{R}{\left(s^2\,C\,L+1\right)\,R+s\,L}$$

(d) O resultado %o10 escreve-se com polinómios, em vez de função racional:

$$\left(R\,L\,C\,s^2+L\,s+R\right)\tilde{V}=R\,\tilde{V}_\mathrm{e}$$

A equação diferencial do sistema é a transformada inversa dessa equação, que por simples inspeção é:

$$R\,L\,C\,\ddot{V}+L\,\dot{V}+R\,V=R\,V_\mathrm{e}$$

Problema 7

O circuito na figura é denominado **filtro passa-baixo**. Escreva a equação que relaciona o sinal de saída com o sinal de entrada. Encontre a função de transferência do sistema e determine o sinal de saída quando o sinal de entrada é o indicado no lado direito da figura. Explique porque se designa este circuito de filtro passa-baixo.



A impedância total é a soma das impedâncias da resistência e do condensador e a transformada da tensão de saída é igual à corrente vezes a impedância do condensador:

$$
\begin{aligned}
z &= R + \frac{1}{C\,s} \\
\tilde{I} &= \frac{\widetilde{V}_\mathrm{e}}{z} = \frac{C\,s\,\widetilde{V}_\mathrm{e}}{R\,C\,s+1} \\
\widetilde{V} &= \frac{\tilde{I}}{C\,s} = \frac{\widetilde{V}_\mathrm{e}}{R\,C\,s+1}
\end{aligned}
$$

Ou seja, a equação diferencial do filtro é:

$$R\,C\,\dot{V} + V = V_\mathrm{e}$$

E a função de transferência é:

$$H = \frac{1}{R\,C\,s+1}$$

Denomina-se passa-baixo, porque $H(s)$ é máxima a baixas frequências ($s \to 0$) e nula a altas frequências ($s \to \infty$).

A expressão do sinal de entrada representado no gráfico, em $t \geq 0$ é:

$$V_\mathrm{e}(t) = V_0(1 - u(t - t_0) = V_0 - V_0\,u(t - t_0)$$

ou seja, a sobreposição linear de um sinal constante, V_0, mais o mesmo sinal, multiplicado por -1 e deslocado no tempo em t_0. Como tal, a resposta do circuito será a resposta ao sinal constante V_0, menos a mesma resposta deslocada no tempo em t_0.

A resposta ao sinal constante V_0 encontra-se multiplicando a sua transformada de Laplace pela função de transferência e calculando a transformada inversa de Laplace:

```
(%i12) v: ratsimp (laplace (V0, t, s)/(R*C*s + 1));
```

(%o12) $\dfrac{V0}{C\,R\,s^2 + s}$

```
(%i13) U: ilt (v, s, t);
```

(%o13) $V0 - V0\,\mathrm{e}^{-\frac{t}{C\,R}}$

O sinal de saída é então:

$$V(t) = V_0 \left(1 - \mathrm{e}^{-\frac{t}{CR}} - u(t - t_0) \left(1 - \mathrm{e}^{-\frac{t-t_0}{CR}}\right)\right)$$

Problema 8

No circuito representado no diagrama, a fonte foi ligada no instante $t = 0$, quando não havia corrente no indutor. Determine a expressão da voltagem na resistência de 3.4 kΩ, em função do tempo t. Com a expressão obtida, confirme as respostas dadas para o problema 11 no capítulo 9.



No domínio da frequência s, o circuito é o seguinte (unidades SI):



O indutor e a resistência de 3.4 kΩ estão em série, podendo ser substituidos por um único dispositivo com impedância $3400 + 0.412\,s$. A transformada da corrente nesse dispositivo será:

```
(%i14) i: ratsimp(5/s/(3400+0.412*s));
```

(%o14) $$\frac{1250}{103\,s^2 + 850000\,s}$$

E a transformada da voltagem na resistência de 3.4 kΩ é:

```
(%i15) v: 3400*i$
```

A voltagem na resistência, em função do tempo, é a transformada inversa de Laplace:

```
(%i16) V: ilt(v,s,t);
```

(%o16) $5-5\,e^{-\frac{850000\,t}{103}}$

A voltagem na resistência de 3.4 kΩ, em função do tempo é:

$$V(t) = 5 - 5\,e^{-\frac{850000}{103}\,t}$$

Finalmente, os resultados do problema 11 do capítulo 9 confirmam-se da forma seguinte:

```
(%i17) dV: diff(V,t)$
(%i18) subst(t=0,V);
(%o18)   0
(%i19) float(subst(t=0,dV));
(%o19)   4.126e+4
(%i20) limit(V,t,inf);
(%o20)   5
```